

ハイゲイン PK 分割位相反転回路

Ayumi's Lab.

2006年1月9日

Revised 2015年10月31日

目次

1 等価回路による解析	2
2 グラフによる解析	5
3 数値例	7
3.1 等価回路による式	7
3.2 グラフによる式	8
3.3 武末氏による式	8
4 シミュレーション	9
4.1 オリジナル	9
4.2 AC バランス後	10
5 ロードライン	11
6 歪率	13

一般に、PK 分割位相反転回路のゲインは 1 以下で、高感度の出力管を用いないと全体のゲインが不足する傾向があります。この欠点を少しでも補おうと、ハイゲイン PK 分割位相反転回路 (図 1) や、ポジネガフィードバック回路などが考案されています。初段と直結した PK 分割位相反転回路では、両相のバランスはもっぱら抵抗の精度により決まるので、AC バランスを取る必要は特にありませんが、ハイゲイン PK 分割位相反転回路では、一般に言われている方法で定数を定めると AC バランスが崩れます。ここでは正確な AC バランスを取るための定数の決め方と、ゲインの計算方法を詳しく見ていくことにします。

ハイゲイン PK 分割位相反転回路では、ブートストラップ用コンデンサ C_b があることにより、低域に時定数を持ちます。この時定数は、

$$T \approx C_b R_d \quad (1)$$

で、一般に R_d は $k\Omega$ 単位ですので、 C_b に $22\mu\text{F}$ 程度を用いれば、カットオフ周波数を数 Hz 以下にでき、この時定数を無視することができます。




図 1: ハイゲイン PK 分割位相反転回路

1 等価回路による解析

中域では、ブートストラップ¹用コンデンサ C_b のインピーダンスは無視できるので、 $R'_{k2} = R_{k2} // R_d$ とおけば、等価回路は図 2 のようになります。




図 2: ハイゲイン PK 分割位相反転の等価回路

等価回路より、

$$-\mu_1 e_i = i_1 r_{p1} + i_1 R_{p1} + (i_1 - i_2) R'_{k2} \quad (2)$$

$$-\mu_2 e_{g2} = i_2 r_{p2} + i_2 R_{p2} + (i_2 - i_1) R'_{k2} \quad (3)$$

$$e_{ok} = (i_1 - i_2) R'_{k2} \quad (4)$$

$$e_{op} = i_2 R_{p2} \quad (5)$$

$$e_{g2} = i_1 R_{p1} \quad (6)$$

という関係が成り立ちます。

式 (6) を式 (3) に代入して整理すると、

$$-i_1(\mu_2 R_{p1} - R'_{k2}) = i_2(r_{p2} + R_{p2} + R'_{k2})$$

$$i_2 = -\frac{\mu_2 R_{p1} - R'_{k2}}{r_{p2} + R_{p2} + R'_{k2}} i_1 \quad (7)$$

$$i_1 - i_2 = i_1 + \frac{\mu_2 R_{p1} - R'_{k2}}{r_{p2} + R_{p2} + R'_{k2}} i_1$$

$$= \frac{\mu_2 R_{p1} + r_{p2} + R_{p2}}{r_{p2} + R_{p2} + R'_{k2}} i_1 \quad (8)$$

これらを式 (4)、(5) に代入して、

$$e_{ok} = \frac{\mu_2 R_{p1} + r_{p2} + R_{p2}}{r_{p2} + R_{p2} + R'_{k2}} i_1 R'_{k2}$$

¹ブートストラップ (bootstrap) は、ブーツの靴ひもであるとの解説が見受けられますが、正しくは、ブーツの靴ひもは bootlace で、bootstrap は編み上げ靴のつまみ皮です。

$$\begin{aligned}
e_{op} &= -\frac{\mu_2 R_{p1} - R'_{k2}}{r_{p2} + R_{p2} + R'_{k2}} i_1 R_{p2} \\
e_o &= e_{g2} + e_{ok} = i_1 R_{p1} + \frac{\mu_2 R_{p1} + r_{p2} + R_{p2}}{r_{p2} + R_{p2} + R'_{k2}} i_1 R'_{k2} \\
&= \left(R_{p1} + \frac{\mu_2 R_{p1} + r_{p2} + R_{p2}}{r_{p2} + R_{p2} + R'_{k2}} R'_{k2} \right) i_1 \\
&= \frac{(r_{p2} + R_{p2} + R'_{k2}) R_{p1} + (\mu_2 R_{p1} + r_{p2} + R_{p2}) R'_{k2}}{r_{p2} + R_{p2} + R'_{k2}} i_1 \\
&= \frac{(r_{p2} + R_{p2})(R_{p1} + R'_{k2}) + (\mu_2 + 1) R_{p1} R'_{k2}}{r_{p2} + R_{p2} + R'_{k2}} i_1
\end{aligned} \tag{9}$$

これより、位相反転段のゲイン A_{2k} , A_{2p} は、

$$\begin{aligned}
A_{2k} &\equiv \frac{e_{ok}}{e_o} = \frac{(\mu_2 R_{p1} + r_{p2} + R_{p2}) R'_{k2}}{(r_{p2} + R_{p2})(R_{p1} + R'_{k2}) + (\mu_2 + 1) R_{p1} R'_{k2}} \\
&= \frac{\frac{\mu_2 R_{p1} + r_{p2} + R_{p2}}{R_{p1} + R'_{k2}} R'_{k2}}{r_{p2} + R_{p2} + (\mu_2 + 1) \frac{R_{p1} R'_{k2}}{R_{p1} + R'_{k2}}}
\end{aligned} \tag{10}$$

$$= \frac{\mu_2 (R_{p1} // R'_{k2}) + \frac{R'_{k2}}{R_{p1} + R'_{k2}} (r_{p2} + R_{p2})}{r_{p2} + R_{p2} + (\mu_2 + 1) (R_{p1} // R'_{k2})} \tag{11}$$

$$\begin{aligned}
A_{2p} &\equiv \frac{e_{op}}{e_o} = -\frac{(\mu_2 R_{p1} - R'_{k2}) R_{p2}}{(r_{p2} + R_{p2})(R_{p1} + R'_{k2}) + (\mu_2 + 1) R_{p1} R'_{k2}} \\
&= -\frac{\frac{\mu_2 R_{p1} - R'_{k2}}{R_{p1} + R'_{k2}} R_{p2}}{r_{p2} + R_{p2} + (\mu_2 + 1) \frac{R_{p1} R'_{k2}}{R_{p1} + R'_{k2}}} \\
&= -\frac{\mu_2 \frac{R_{p1} R_{p2}}{R_{p1} + R'_{k2}} - \frac{R'_{k2}}{R_{p1} + R'_{k2}} R_{p2}}{r_{p2} + R_{p2} + (\mu_2 + 1) (R_{p1} // R'_{k2})}
\end{aligned} \tag{12}$$

プレート側の負荷抵抗には i_2 しか流れないのに対して、カソード側の負荷抵抗には、前段の出力電流も加わった $i_2 - i_1$ の電流が流れるので、プレート側とカソード側の抵抗が等しいと、必ずカソード側の出力が大きくなります。

両相がバランスするためには、いずれかの負荷抵抗を変える必要があります。両相がバランスする条件は、式 (10), (12) の絶対値が等しいということであり、

$$(\mu_2 R_{p1} + r_{p2} + R_{p2}) R'_{k2} = (\mu_2 R_{p1} - R'_{k2}) R_{p2}$$

プレート側の負荷抵抗を変えるほうが影響が少ないので、 R_{p2} を変更することにします。上式を R_{p2} について解くと、

$$R_{p2} = \frac{\mu_2 R_{p1} + r_{p2}}{\mu_2 R_{p1} - 2R'_{k2}} R'_{k2} \tag{14}$$

となり、プレート側の負荷抵抗値をこの値にすれば、両相のバランスが取れます。プレート電圧が変わり、三定数も変わるので、場合によっては R_{p2} を再計算する必要があります。

位相反転段のプレート特性図にカソード側のロードラインを引く場合の実質的なカソード抵抗の値を R_{k2}^* とすれば、

$$i_2 R_{k2}^* = (i_2 - i_1) R'_{k2}$$

という関係が成り立ちます．この式の右辺はプレート側に i_2 が流れたときにカソードに現れる電圧を表しています．プレート特性図は i_2 に関して描かれているため， i_2 に対してこの電圧が生じるようなロードラインを引けばよいのです．

$$\begin{aligned} R_{k2}^* &= \frac{i_2 - i_1}{i_2} R'_{k2} \\ &= \frac{\mu_2 R_{p1} + r_{p2} + R_{p2}}{\mu_2 R_{p1} - R'_{k2}} R'_{k2} \end{aligned} \quad (15)$$

となります．

ここで，通常の PK 分割位相反転回路との比較を行ってみます． $R_{p2} = R'_{k2} = R_L$ とおくと，式 (11), (13) は，次のようになり，

$$\begin{aligned} A_{2k} &= \frac{\mu_2(R_{p1} // R_L) + \frac{R_L}{R_{p1} + R_L}(r_{p2} + R_L)}{r_{p2} + R_L + (\mu_2 + 1)(R_{p1} // R_L)} \\ A_{2p} &= -\frac{\mu_2(R_{p1} // R_L) - \frac{R_L}{R_{p1} + R_L} R_L}{r_{p2} + R_L + (\mu_2 + 1)(R_{p1} // R_L)} \end{aligned}$$

分子の第 2 項のみが異なることがわかります．一方，通常の PK 分割位相反転回路のゲインは，

$$A_2 = \pm \frac{\mu_2 R_L}{r_{p2} + (\mu_2 + 2)R_L}$$

です．両方の式を比較すると，ハイゲイン PK 分割位相反転回路では，分子の第 1 項と，分母に R_{p1} が並列になって入っているため，ややゲインが下がります．それに加え，カソード側はゲインが大きくなる要素が，プレート側はゲインが小さくなる要素がありますが， R_{p1} が大きくなれば $R_L / (R_{p1} + R_L)$ の値が小さくなるので，その影響は小さくなります．とは言うものの，ブートストラップによって実質的な R_{p1} は非常に大きくなるので， R_{p1} にはそれほど大きな値の抵抗を使わないので，アンバランスは相当の大きさで残ることになります．したがって，上述の方法などでバランスを取ることは必須と言えます．

次に初段のゲインを求めます．式 (9) のカッコ内を R'_{p1} とおくと，

$$R'_{p1} \equiv R_{p1} + \frac{\mu_2 R_{p1} + r_{p2} + R_{p2}}{r_{p2} + R_{p2} + R'_{k2}} R'_{k2}$$

一方，式 (10) より，

$$\begin{aligned} 1 - A_{2k} &= 1 - \frac{(\mu_2 R_{p1} + r_{p2} + R_{p2})R'_{k2}}{(r_{p2} + R_{p2})(R_{p1} + R'_{k2}) + (\mu_2 + 1)R_{p1}R'_{k2}} \\ &= 1 - \frac{(\mu_2 R_{p1} + r_{p2} + R_{p2})R'_{k2}}{(r_{p2} + R_{p2} + R'_{k2})R_{p1} + (\mu_2 R_{p1} + r_{p2} + R_{p2})R'_{k2}} \\ &= \frac{(r_{p2} + R_{p2} + R'_{k2})R_{p1}}{(r_{p2} + R_{p2} + R'_{k2})R_{p1} + (\mu_2 R_{p1} + r_{p2} + R_{p2})R'_{k2}} \end{aligned}$$

であるから，

$$\begin{aligned} \frac{R_{p1}}{1 - A_{2k}} &= R_{p1} \cdot \frac{(r_{p2} + R_{p2} + R'_{k2})R_{p1} + (\mu_2 R_{p1} + r_{p2} + R_{p2})R'_{k2}}{(r_{p2} + R_{p2} + R'_{k2})R_{p1}} \\ &= R_{p1} + \frac{\mu_2 R_{p1} + r_{p2} + R_{p2}}{(r_{p2} + R_{p2} + R'_{k2})R_{p1}} R'_{k2} = R'_{p1} \end{aligned}$$

これと式 (8) を使って式 (2) を整理すると ,

$$\begin{aligned}
 -\mu_1 e_i &= i_1(r_{p1} + R_{p1}) + (i_1 - i_2)R'_{k2} \\
 &= i_1(r_{p1} + R_{p1}) + \frac{\mu_2 R_{p1} + r_{p2} + R_{p2}}{r_{p2} + R_{p2} + R'_{k2}} i_1 R'_{k2} \\
 &= i_1 \left(r_{p1} + R_{p1} + \frac{\mu_2 R_{p1} + r_{p2} + R_{p2}}{r_{p2} + R_{p2} + R'_{k2}} R'_{k2} \right) \\
 &= i_1(r_{p1} + R'_{p1}) \\
 i_1 &= -\frac{\mu_1 e_i}{r_{p1} + R'_{p1}} \tag{16}
 \end{aligned}$$

初段のゲイン A_1 は ,

$$A_1 \equiv \frac{e_o}{e_i} = \frac{i_1 R'_{p1}}{e_i} = -\frac{\mu_1 R'_{p1}}{r_{p1} + R'_{p1}} \tag{17}$$

となります . この式から , 初段の等価的な負荷抵抗は R'_{p1} である , すなわち R_{p1} を $1/(1 - A_{2k})$ 倍したものと いえます .

2 グラフによる解析

初段の動作点を O とします . 動作点のプレート電圧を E_{p0} , グリッド電圧を E_{g0} , プレート電流を I_{p0} とします . ここに入力電圧 e_i を加えた場合を考えます . プレート特性図を図 3 に示します .




図 3: ハイゲイン PK 分割位相反転初段のプレート特性図

初段の出力電圧を e_o , 位相反転のカソード側出力を e_{ok} とします . 位相反転段のゲインを A_2 とすれば ,

$$e_{ok} = A_2 e_o \tag{18}$$

という関係が成り立ちます .

この電圧が , ブートストラップ用のコンデンサを通じてプレート負荷抵抗に加えられるため , R_{p1} の下側の電圧は , 無信号時よりも $|e_{ok}|$ だけ低くなります . したがって , ロードラインは左側に $|e_{ok}|$

だけずれた状態になります。このずれたロードラインと、 $E_g = E_{g0} + e_i$ のプレート特性曲線が交わったところ (点 C) が、信号 e_i を加えたときのプレート電圧とプレート電流になります。

この時のプレート電流の変化を i_p とすると、図より、

$$DA = i_p R_{p1} \quad (19)$$

$$BD = i_p r_{p1} \quad (20)$$

となります。また、図より、

$$e_o + DA = A_2 e_o \quad (21)$$

$$-\mu e_i + BD = e_o \quad (22)$$

が成り立つので、

$$DA = -(1 - A_2)e_o \quad (23)$$

$$BD = \mu e_i + e_o \quad (24)$$

これらを式 (20), (19) に代入すると、

$$-(1 - A_2)e_o = i_p R_{p1} \quad (25)$$

$$\mu e_i + e_o = i_p r_{p1} \quad (26)$$

となり、これらから i_p を消去して、

$$\begin{aligned} \frac{-(1 - A_2)e_o}{R_{p1}} &= \frac{\mu e_i + e_o}{r_{p1}} \\ \left(\frac{1 - A_2}{R_{p1}} + \frac{1}{r_{p1}}\right)e_o &= \frac{-\mu e_i}{r_{p1}} \\ \frac{(1 - A_2)r_{p1} + R_{p1}}{r_{p1}R_{p1}}e_o &= -\frac{\mu e_i}{r_{p1}} \\ \{(1 - A_2)r_{p1} + R_{p1}\}e_o &= -\mu R_{p1}e_i \\ e_o &= \frac{-\mu R_{p1}}{(1 - A_2)r_{p1} + R_{p1}}e_i \\ A_1 &= \frac{-\mu R_{p1}}{(1 - A_2)r_{p1} + R_{p1}} \end{aligned} \quad (27)$$

実質的なロードライン OC の負荷抵抗値 R'_{p1} は、

$$\begin{aligned} i_p &= \frac{DA}{R_{p1}} = \frac{DO}{R'_{p1}} \\ \frac{-(1 - A_2)e_o}{R_{p1}} &= \frac{-e_o}{R'_{p1}} \\ R'_{p1} &= \frac{R_{p1}}{1 - A_2} \end{aligned} \quad (28)$$

となります。一般に A_2 は 0.9~0.95 ですから、 R'_{p1} は R_{p1} の 10 倍から 20 倍になり、ゲインは初段の μ とほとんど等しくなります。このように、ハイゲイン PK 分割位相反転回路では、ブートストラップにより初段の実質的な負荷抵抗が非常に高くなることで、大きなゲインを得ています。負荷抵抗が高くなることで、歪率も下がり、最大出力電圧も大きくなります。そのかわり、負荷抵抗に

よる出力インピーダンスの低下が期待できないので、初段に多極管を使った場合は、高域特性が悪くなります。

次に、位相反転段のゲインの求め方ですが、正確にはブートストラップコンデンサを経由して初段から流れてくる電流 (i_p) も考慮しなければならないのですが、グラフでわかるように、みかけのインピーダンスが非常に高くなっており、仮に $R_{p1} = R'_{k2}$ としても、 i_p は i_{p2} よりも一桁以上小さいため、無視してゲインを計算してもよいでしょう。

3 数値例

ここでは、[1] の 6BQ5 プッシュプルアンプを検証します。位相反転段までの回路図を図 4 に示します。




図 4: ハイゲイン PK 分割位相反転回路の例

V1 の動作点は、 $E_{p10} = 98.20391 \text{ V}$, $E_{g10} = -2.32425 \text{ V}$, $I_{p10} = 7.043193 \text{ mA}$, $\mu_1 = 30.07848$, $r_{p1} = 3646.169 \Omega$, $g_{m1} = 8.249339 \text{ mS}$ で、V2 の動作点は、 $E_{p20} = 104.38063 \text{ V}$, $E_{g20} = -2.55142 \text{ V}$, $I_{p20} = 6.871972 \text{ mA}$, $\mu_2 = 29.85942$, $r_{p2} = 3720.371 \Omega$, $g_{m2} = 8.025924 \text{ mS}$ です。計算に必要な抵抗値は、 $R_{p1} = 7.5 \text{ k}\Omega$, $R'_{p2} = R_{p2} // R_{g3} = 7.5 // 330 \text{ k}\Omega = 7.333 \text{ k}\Omega$, $R'_{k2} = R_{k2} // R_d // R_{g4} = 15 // 15 // 330 \text{ k}\Omega = 7.333 \text{ k}\Omega$ です。

3.1 等価回路による式

$$\begin{aligned}
 A_{2k} &= \frac{(\mu_2 R_{p1} + r_{p2} + R_{p2}) R'_{k2}}{(r_{p2} + R_{p2})(R_{p1} + R'_{k2}) + (\mu_2 + 1) R_{p1} R'_{k2}} \\
 &= \frac{(29.86 \cdot 7.5 + 3.720 + 7.333) \cdot 7.333}{(3.720 + 7.333)(7.5 + 7.333) + (29.86 + 1) \cdot 7.5 \cdot 7.333} = 0.9259 \\
 A_{2p} &= -\frac{(\mu_2 R_{p1} - R'_{k2}) R_{p2}}{(r_{p2} + R_{p2})(R_{p1} + R'_{k2}) + (\mu_2 + 1) R_{p1} R'_{k2}} \\
 &= -\frac{(29.86 \cdot 7.5 - 7.333) \cdot 7.333}{(3.720 + 7.333)(7.5 + 7.333) + (29.86 + 1) \cdot 7.5 \cdot 7.333} = -0.8535 \\
 R'_{p1} &= \frac{R_{p1}}{1 - A_{2k}} = \frac{7.5}{1 - 0.9259} = 101.2 \text{ [k}\Omega\text{]} \\
 A_1 &= -\frac{\mu_1 R'_{p1}}{r_{p1} + R'_{p1}} = -\frac{30.08 \cdot 101.2}{3.646 + 101.2} = 29.03
 \end{aligned}$$

両相がバランスするための R'_{p2} の値は，

$$R'_{p2} = \frac{\mu_2 R_{p1} + r_{p2}}{\mu_2 R_{p1} - 2R'_{k2}} R'_{k2} = \frac{29.86 \cdot 7.5 + 3.720}{29.86 \cdot 7.5 - 2 \cdot 7.333} \cdot 7.333 = 7.978 \text{ [k}\Omega\text{]}$$

となりますが，これは R_{g3} と並列になった値ですので， R_{p2} 単独の値は，

$$R_{p2} = \frac{1}{\frac{1}{7.978} - \frac{1}{330}} = 8.175 \text{ [k}\Omega\text{]}$$

となり，動作点を変化させないためには，

$$6.872 \cdot (8.175 - 7.5) = 4.64 \text{ [V]}$$

だけ電源電圧を高くする必要があります．

3.2 グラフによる式

PK 分割段のゲインの計算は，通常どおりに行ないました．

$$\begin{aligned} A_2 &= \mu_2 \frac{R'_{p2}}{r_{p2} + (\mu_2 + 2)R'_{p2}} \\ &= 29.86 \frac{7.333}{3.720 + (29.86 + 2) \cdot 7.333} = 0.9225 \\ A_1 &= -\frac{\mu R_{p1}}{(1 - A_2)r_{p1} + R_{p1}} \\ &= -\frac{30.08 \cdot 7.5}{(1 - 0.9225) \cdot 3.646 + 7.5} = -28.99 \end{aligned}$$

等価回路による結果と多少違っていますが，実用的には十分な精度でしょう．等価回路による式から求めた A_{2k} を用いれば，結果は等価回路のものと一致します．

3.3 武末氏による式

武末氏は，[2, p. 200] で，ブートストラップがある場合の増幅度 A' を

$$A' = A \cdot \left(1 + \frac{E_c}{E_o} \cdot \frac{r_p}{R_p + r_p}\right) \quad (29)$$

としています．ここで， E_o は初段の出力電圧， E_c はブートストラップの電圧ですから， $E_c/E_o = A_{2k}$ です．この式によりゲインを計算すると，

$$\begin{aligned} A' &= -\mu_1 \frac{R_{p1}}{r_{p1} + R_{p1}} \cdot \left(1 + A_2 \cdot \frac{r_{p1}}{R_{p1} + r_{p1}}\right) \\ &= -30.08 \frac{7.5}{3.646 + 7.5} \cdot \left(1 + 0.9225 \cdot \frac{3.646}{7.5 + 3.646}\right) \\ &= -20.240 \cdot 1.3029 \\ &= -26.37 \end{aligned}$$

となり，いくぶん低めに算出されます．

この式は，ブートストラップの電圧が初段の負荷抵抗と内部抵抗で分圧された値が，単にブートストラップのない場合の出力に加わったものです．実際には，ブートストラップがあると初段の動作が大きく変わってしまうので，この式では正しい増幅度が求められないということです．

4 シミュレーション

4.1 オリジナル

```
1 High gain PK phase splitter (Original)
2 .INCLUDE ECC88.inc
3
4 VI 1 0 DC 0 AC 1 SIN(0 0.1 1kHz)
5 R1 1 0 100k
6 XV1 2 1 3 ECC88
7 R2 3 4 300
8 C1 3 0 470u
9 R3 4 0 30
10 VIP1 5 2 0
11 R4 5 6 7.5k
12 R5 6 7 15k
13 VB 7 0 259V
14 XV2 8 5 9 ECC88
15 R6 9 0 15k
16 C2 9 6 220u
17 VIP2 10 8 0
18 R7 7 10 7.5k
19 C3 10 11 0.68u
20 R8 11 0 330k
21 C4 9 12 0.68u
22 R9 12 0 330k
23
24 .NODESET V(2)=100.5V V(6)=153.3V V(8)=207.5V V(9)=103V
25
26 .control
27 op
28 print v(2) v(3) v(2,3) v(1,3)
29 print i(vip1)
30 print v(8) v(5) v(9) v(8,9) v(5,9)
31 print i(vip2)
32 print v(6) v(6,9)
33 ac dec 30 1k 1k
34 print abs(v(2)) db(v(2))
35 print abs(v(8)/v(2)) db(v(8)/v(2))
36 print abs(v(9)/v(2)) db(v(9)/v(2))
37 .endc
38 .END
```

出力結果

```
1
2 Circuit: high gain pk phase splitter (original)
3
4 v(2) = 1.005281e+02
5 v(3) = 2.324254e+00
6 v(2,3) = 9.820389e+01
7 v(1,3) = -2.32425e+00
8 i(vip1) = 7.043194e-03
9 v(8) = 2.074606e+02
10 v(5) = 1.005281e+02
11 v(9) = 1.030788e+02
12 v(8,9) = 1.043819e+02
13 v(5,9) = -2.55061e+00
14 i(vip2) = 6.871917e-03
15 v(6) = 1.533521e+02
16 v(6,9) = 5.027334e+01
17
18 abs(v(2)) = 2.903273e+01 # A1
19 db(v(2)) = 2.925776e+01
20 abs(v(8)/v(2)) = 8.534674e-01 # A2p
21 db(v(8)/v(2)) = -1.37626e+00
22 abs(v(9)/v(2)) = 9.259105e-01 # A2k
```

```
23 db(v(9)/v(2)) = -6.68620e-01
```

ゲインはプレート側 0.8535 倍，カソード側 0.9259 倍となり，0.71 dB のアンバランスがあります。

4.2 AC バランス後

```
1 High gain PK phase splitter (Case 2--AC balanced)
2 .INCLUDE ECC88.inc
3
4 VI 1 0 DC 0 AC 1 SIN(0 0.1 1kHz)
5 R1 1 0 100k
6 XV1 2 1 3 ECC88
7 R2 3 4 300
8 C1 3 0 470u
9 R3 4 0 30
10 VIP1 5 2 0
11 R4 5 6 7.5k
12 R5 6 7 15k
13 VB 7 0 259V
14 XV2 8 5 9 ECC88
15 R6 9 0 15k
16 C2 9 6 220u
17 VIP2 10 8 0
18 R7 77 10 8.175267k
19 C3 10 11 0.68u
20 R8 11 0 330k
21 C4 9 12 0.68u
22 R9 12 0 330k
23 VBB 77 0 263.6404
24
25 .NODESET V(2)=100.5V V(6)=153.3V V(8)=207.5V V(9)=103V
26
27 .control
28 op
29 print v(2) v(3) v(2,3) v(1,3)
30 print i(vip1)
31 print v(8) v(5) v(9) v(8,9) v(5,9)
32 print i(vip2)
33 print v(6) v(6,9)
34 ac dec 30 1k 1k
35 print abs(v(2)) db(v(2))
36 print abs(v(8)/v(2)) db(v(8)/v(2))
37 print abs(v(9)/v(2)) db(v(9)/v(2))
38 .endc
39 .END
```

出力結果

```
1
2 Circuit: high gain pk phase splitter (case 2--ac balanced)
3
4 v(2) = 1.005281e+02
5 v(3) = 2.324254e+00
6 v(2,3) = 9.820389e+01
7 v(1,3) = -2.32425e+00
8 i(vip1) = 7.043194e-03
9 v(8) = 2.074606e+02
10 v(5) = 1.005281e+02
11 v(9) = 1.030788e+02
12 v(8,9) = 1.043819e+02
13 v(5,9) = -2.55062e+00
14 i(vip2) = 6.871917e-03
15 v(6) = 1.533521e+02
16 v(6,9) = 5.027334e+01
17
```

```

18 abs(v(2)) = 2.900272e+01 # A1
19 db(v(2)) = 2.924878e+01
20 abs(v(8)/v(2)) = 9.237088e-01 # A2p
21 db(v(8)/v(2)) = -6.89299e-01
22 abs(v(9)/v(2)) = 9.237058e-01 # A2k
23 db(v(9)/v(2)) = -6.89326e-01

```

ゲインはプレート側 0.9237 倍，カソード側 0.9237 倍となり，AC バランスが取れていることがわかります。

5 ロードライン

オリジナルの初段のロードラインを図 5 に示します。点線は等価回路により求めたロードライン




図 5: ハイゲイン PK 分割位相反転初段のロードライン

です。実質的な負荷抵抗が非常に大きくなるので，プレート電流をもう少し絞っても低歪みを維持できるでしょう。

オリジナルの位相反転段のロードラインを図 6 に示します。青い線が V2 のロードライン，緑色の線がカソードの電圧，オレンジ色の線が対グラウンドのプレート電圧です。最大出力電圧をさらに増やすためには，パイアスを深くし，プレート電流を下げる必要があります。ご覧のようにプレート側とカソード側でロードラインの傾きが違うことがわかります。2.6 V_{p-p} の入力に対して，プレート側は，

$$240.1 - 175.2 = 64.9 [V_{p-p}]$$

カソード側は，

$$137.8 - 67.4 = 70.4 [V_{p-p}]$$

の出力が得られます。

AC バランス後の位相反転段のロードラインを図 7 に示します。2.6 V_{p-p} の入力に対して，プレート側は，

$$242.7 - 172.6 = 70.1 [V_{p-p}]$$




図 6: ハイゲイン PK 分割位相反転段のロードライン




図 7: AC バランス後のハイゲイン PK 分割位相反転段のロードライン

カソード側は，

$$137.7 - 67.6 = 70.1 [V_{p-p}]$$

の出力が得られ，バランスしていることがわかります．

6 歪率

同じ動作点で通常の PK 分割を構成し，同じ入力電圧に対する出力電圧と歪率を比較してみました．

項目	ハイゲイン	PK 分割
入力電圧 (V_p)	1.2	1.2
初段歪率 (%)	1.000	3.551
プレート側出力電圧 (V_p)	33.64	23.34
プレート側歪率 (%)	0.288	3.293
カソード側出力電圧 (V_p)	33.64	23.34
カソード側歪率 (%)	0.658	3.293

通常の PK 分割でも，ハイゲイン PK 分割でも，初段の歪率よりも位相反転段の歪率が低くなっています．位相反転段は強力な負帰還がかかっており，ゲインが 1 未満と低くなっていますが，残った歪みは前段の歪みを打ち消す方向となっています．

ハイゲイン PK 分割位相反転回路では，カソード側の歪率がプレート側よりも悪くなっています．これは，カソードの抵抗に初段の電流も流れるため，歪の打ち消しがプレート側よりも弱くなっているためです．カソード側で比較しても，通常の PK 分割と比べて歪率が約 1/5 となり，なおかつ出力電圧も高くとれるので，通常の PK 分割を使うメリットはほとんどないと言ってよいでしょう．

参考文献

- [1] 黒川達夫. 現代真空管アンプ 25 選. 誠文堂新光社, 1998.
- [2] 武末数馬. パワーアンプの設計と製作 (上). ラジオ技術社, 第 7 版, 1974.